CS344 Lab 13 – Gauss Elimination

Assigned: 11/13/09
Points: 20

Purpose

The purpose of this lab is to implement and illustrate the Gauss elimination technique in FORTRAN.

Process

Your program will read in a series of equations and then use the Gauss elimination technique to solve them. Use the code and the data1.dat file listed below as the starter program. Execute the program to verify that it works correctly using the data file provided.

Modify the program to do the following:

- Eliminate the use of the GOTO statement.
- Add a Print subroutine as discussed on page 484
- Integrate the Print subroutine into the application to illustrate the execution of the Gauss elimination technique as shown on page 479

Run the program against the other data files that will be provided.

Turn in your work as you did for the previous labs.

```
      PROGRAM GAUSS

      IMPLICIT NONE
      INTEGER DROWS, DCOLS, N, I, J, PIVOT
      REAL A(25,26), SOLN(25)
      LOGICAL ERROR

      DROWS = 25
      DCOLS = 26
      OPEN (UNIT=10,FILE='DATA1.DAT',STATUS='OLD')
      READ(10,*)N
      DO 5 I=1,N
           READ(10,*) (A(I,J),J=1,N+1)
    5 CONTINUE

      PIVOT = 1
      ERROR = .FALSE.

   10 IF (PIVOT.LT.N.AND..NOT.ERROR) THEN
           CALL ORDER(A,DROWS,DCOLS,N,PIVOT,ERROR)
           IF(.NOT.ERROR) THEN
                CALL ELIM(A,DROWS,DCOLS,N,PIVOT)
                PIVOT = PIVOT + 1
           ENDIF
           GOTO 10
      ENDIF

      IF(ERROR) THEN
           PRINT *,'NO UNIQUE SOLUTION EXISTS'
      ELSE
           CALL BACKSB(A,DROWS,DCOLS,N,SOLN)
           PRINT *, 'SOLUTION WRITTEN TO DATA FILE'
           OPEN (UNIT=11,FILE='ACTUATOR.DAT',STATUS='NEW');
           WRITE(11,*) N
           DO 20 I=1,N
                WRITE(11,*) SOLN(I)
   20      CONTINUE
      ENDIF

      END
*****************************************************************
*
      SUBROUTINE ORDER(A,DROWS,DCOLS,N,PIVOT,ERROR)
      INTEGER DROWS, DCOLS, N, ROW, RMAX, PIVOT, K
      REAL A(DROWS,DCOLS), TEMP
      LOGICAL ERROR

      RMAX = PIVOT
      DO 10 ROW=PIVOT+1,N
           IF(ABS(A(ROW,PIVOT)).GT.ABS(A(RMAX,PIVOT))) RMAX = ROW
   10 CONTINUE

      IF(ABS(A(RMAX,PIVOT)).LT.1.0E-05) THEN
           ERROR = .TRUE.
      ELSE
           IF(RMAX.NE.PIVOT) THEN
                DO 20 K=1,N+1
```

```
                    TEMP = A(RMAX,K)
                    A(RMAX,K) = A(PIVOT,K)
                    A(PIVOT,K) = TEMP
   20           CONTINUE
             ENDIF
          ENDIF

          RETURN
          END
*****************************************************************
*
          SUBROUTINE ELIM(A,DROWS,DCOLS,N,PIVOT)
          INTEGER DROWS, DCOLS, N, PIVOT, ROW, COL
          REAL A(DROWS,DCOLS), FACTOR

          DO 10 ROW=PIVOT+1,N
               FACTOR = A(ROW,PIVOT)/A(PIVOT,PIVOT)
               A(ROW,PIVOT) = 0.0
               DO 5 COL=PIVOT+1,N+1
                   A(ROW,COL) = A(ROW,COL) - A(PIVOT,COL)*FACTOR
    5          CONTINUE
   10     CONTINUE

          RETURN
          END
*****************************************************************
*
          SUBROUTINE BACKSB(A,DROWS,DCOLS,N,SOLN)
          INTEGER DROWS, DCOLS, N, ROW, COL
          REAL A(DROWS,DCOLS), SOLN(DROWS)

          DO 20 ROW=N,1,-1
               DO 10 COL=N,ROW+1,-1
                   A(ROW,N+1) = A(ROW,N+1) - SOLN(COL)*A(ROW,COL)
   10          CONTINUE
               SOLN(ROW) = A(ROW,N+1)/A(ROW,ROW)
   20     CONTINUE

          RETURN
          END
```

data1.dat

```
3
2  3  -1  1
3  5  2  8
1 -2 -3 -1
```